

```
$oninline
/*
GAMS program used to estimate technical efficiency.
```

```
Author: John B. Walden
        NMFS/NEFSC
        166 Water St.
        Woods Hole, MA 02543
        (508) 495-2355
        John.Walden@Noaa.Gov
```

```
This version will estimate technical efficiency
based on an input oriented DEA model, with
variable returns to scale. Models can
be found in Fare, Grosskopf and Lovell (1994)*/
```

```
/* The following line turns off listing of
some elements in the GAMS listing file*/
```

```
$OFFSYMLIST OFFSYMXPREF OFFFUELLIST OFFFUELXPREF
```

```
OPTION SOLPRINT=OFF, SYSOUT=OFF, LIMROW=0, LIMCOL=0;
```

```
/*NEXT DEFINE INPUTS AND OUTPUTS*/
```

```
SET INOUT /x1*x6, y1*y3/
```

```
        OUTPUT(INOUT) /y1*y3/
```

```
        INPUT(INOUT) /x1*x6/
```

```
OBS /1*2000/
```

```
SUBOBS(OBS) /1*201/
```

```
ACTOBS(OBS)
```

```
;
```

```
/*Next, define an alias for the set SUBOBS */
```

```
alias (subobs, subobs1);
```

```
$OFFLISTING
```

```
TABLE ACT(OBS,INOUT) INPUT OUTPUT TABLE
```

```
$ondelim
```

```
$INCLUDE "data2.csv"
```

```
$offdelim
```

```
$ONLISTING
```

```
VARIABLES
```

```
lambda      efficiency score
weight(obs) weights;
```

```
POSITIVE Variable weight;
```

EQUATIONS

```
CONSTR1(OUTPUT, OBS)  DEA constraint for each output
CONSTR2(Input,OBS)    DEA Constraint for each input and calculating TE
CONSTR3                DEA Constraint for imposing VRS;
```

```
CONSTR1(OUTPUT, ACTOBS).. SUM(SUBOBS, WEIGHT(SUBOBS)*ACT(SUBOBS,OUTPUT)) =G= ACT(ACTOBS,
OUTPUT);
```

```
CONSTR2(INPUT, ACTOBS).. SUM(SUBOBS, WEIGHT(SUBOBS)*ACT(SUBOBS,INPUT)) =L=
LAMBDA*ACT(ACTOBS,INPUT);
```

```
CONSTR3..                SUM(SUBOBS, WEIGHT(SUBOBS)) =E= 1;
```

```
MODEL TE /all/;
```

```
te.solprint=2;           /*Turn off writing results to solutions file*/
te.solverlink=2;        /*Keep model in memory. Improves solution time*/
```

```
file primal /teinp_vrs.txt/;
```

```
parameter score1(obs) efficiency scores;
```

```
LOOP(SUBOBS1,
```

```
    ACTOBS(OBS)=NO;
    ACTOBS(SUBOBS1)=YES;
    SOLVE TE minimizing lambda using LP;
```

```
    score1(SUBOBS1) = lambda.l;
```

```
    put primal;
```

```
    if ((te.modelstat eq 1 and te.solvestat eq 1),
```

```
        put @1, subobs1.tl, @10, "optimal", @25, "normal completion"/
```

```
    else
```

```
        put @1, subobs1.tl, @10, te.modelstat:>2:0, @25, te.solvestat:>2.0/
    ); /*end if*/
```

```
); /*end loop*/
```

```
/*results are now printed out to the file 'te_inp_vrs.csv'.
There are other ways to place results in an external file. Refer
to the GAMS manual for more information. Since this program is run on
a Linux system, I use the put utility which comes with GAMS*/
```

```
file sr /te_inp_vrs.csv/;
```

```
sr.pc=5;
```

```
put sr;
```

```
put 'obs', 'lambda';
put /;
loop(subobs, put subobs.tl, score1(subobs):4;
  put /;

);

putclose; /*closes external files*/
```